
1 Correction to LTE-UTRA testcases
1.1 Change 1
	Object name 
	fl_TC_9_2_3_2_1a_Body

	Reason for change
	According to the prose, the IE T3412 value in tracking area update accept message should be omit, and the current TTCN implementation includes it

	Summary of change
	Introduced a new template “cds_TAU_Accept_T3412” and set the IE T3412 to omit.

	Source of change
	

	Module
	/LTE_IRAT/9_2/NAS_TrackingArea_UG.ttcn


Before:

	...

function fl_TC_9_2_3_2_1a_Body() runs on EUTRA_PTC
  {
...
    //@siclog "Step 1" siclog@
    // Set cell type of Cell A to the "non-Suitable cell"
    // Set cell type of Cell B to the "Serving cell".
    v_CellPowerList := {
      cs_CellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE),
      cs_CellPower(eutra_CellB, tsc_ServingCellRS_EPRE)
    };
  ...
    //@siclog "Step 3" siclog@
    //The SS sends TRACKING AREA UPDATE ACCEPT message including with LAI set to LAI-2, including
    //         a new TMSI(TMSI-2) as MS identity
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_TAU_Accept(v_UpdateType,
                                                                  omit,
                                                                  v_TaiList2,
                                                                  v_EpsBearerCtxtStatus,
                                                                  v_LAI_2,
                                                                  v_MsId,
                                                                  v_AdditionalUpdateResult))));
    //@siclog "Step 4" siclog@
    // Check: Does the UE send TRACKING AREA UPDATE COMPLETE message.
    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    ...
    //@siclog "Step 18" siclog@
    // The SS sends TRACKING AREA UPDATE ACCEPT message with LAI set to LAI-1, not including MS identity.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_TAU_Accept(v_UpdateType,
                                                                  omit,
                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN, { v_TAC1 }),
                                                                  v_EpsBearerCtxtStatus,
                                                                  v_LAI_1,
                                                                  omit,
                                                                  v_AdditionalUpdateResult))));
    //@siclog "Step 19" siclog@
    // The SS releases the RRC connection
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    ...
    //@siclog "Step 30" siclog@
    //  The SS sends TRACKING AREA UPDATE ACCEPT message with LAI set to LAI-2, including
    //         the IMSI as MS identity.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_TAU_Accept(v_UpdateType,
                                                                  omit,
                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN_B, { v_TAC2 }),
                                                                  v_EpsBearerCtxtStatus,
                                                                  v_LAI_2,
                                                                  v_MsId_IMSI,
                                                                  v_AdditionalUpdateResult))));
    // EXCEPTION: Step 30a is optional for Release 8 to Release 10 UE implementation, from release 11 onwards, UE shall not send this message.
    //@siclog "Step 30a" siclog@
    // The UE sends a TRACKING AREA UPDATE COMPLETE message.  @sic R5-125738 sic@
    v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease();
    if ((v_Rel == rel8) or (v_Rel == rel9) or (v_Rel == rel10)) {
      t_Wait2.start; // 2 sec considered sufficient
      alt {
        [] SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)))
          {
            t_Wait2.stop;
          };
        [] t_Wait2.timeout {};
      }
    }
    //@siclog "Step 31" siclog@
    // The SS releases the RRC connection
    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);
    // The following messages are sent and shall be received on Cell X
    //@siclog "Step 32" siclog@
    // Set cell type of Cell B to the "non-Suitable cell". Set cell type of Cell A to off
    v_CellPowerList := {
      cs_CellPower(eutra_CellA, tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_CellB, tsc_NonSuitableCellRS_EPRE)
    };
    f_EUTRA_SetCellPowerList(v_CellPowerList);   // @sic R5s110726 additional changes for change 2 sic@
    f_EUTRA_SendAuthParameters(eutra_CellB, UTRAN); // @sic R5s110685 sic@
    // wait for trigger from UTRAN side to finish TC
    f_IRAT_WaitForCoOrd_Trigger(UTRAN);
  }// end function fl_TC_9_2_3_2_1a_Body
...


After:

	...

function fl_TC_9_2_3_2_1a_Body() runs on EUTRA_PTC
  {
    var template (value) CellPowerList_Type v_CellPowerList;
    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var GutiParameters_Type v_Guti2_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellB);
    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);
    var NAS_PlmnId v_PLMN_B := f_Asn2Nas_PlmnId(v_Guti2_Params.PLMN_Identity);
    var NAS_Lac v_Lac1 := f_EUTRA_CellInfo_GetLocationAreaCode(eutra_CellA);
    var NAS_Lac v_Lac2 := '0002'O;
    var NAS_Lac v_Lac3 := '0003'O;
    var NAS_Lac v_Lac4 := '0004'O;
    var NAS_Tac v_TAC1  := '0001'O;
    var NAS_Tac v_TAC2  := '0002'O;
    var template (value) LocAreaId v_LAI_1 := cs_LAI('13'O, v_PLMN, v_Lac1);
    var template (value) LocAreaId v_LAI_2 := cs_LAI('13'O, v_PLMN, v_Lac2);
    var template (value) LocAreaId v_LAI_3 := cs_LAI('13'O, v_PLMN, v_Lac3);
    var template (value) LocAreaId v_LAI_4 := cs_LAI('13'O, v_PLMN, v_Lac4);
    var template (omit) MobileIdentity v_MsId := f_GetMSId(Combined); // This will use px_TMSI_Def
    var template (value) TrackingAreaIdList v_TaiList2 := cds_TAIListNonConsecutive_tlv(v_PLMN_B, { v_TAC2 });
    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;
    var template (omit) AdditionalUpdateResult v_AdditionalUpdateResult;
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType(Combined); // @sic R5s110176 sic@
    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(Combined);
    var NAS_MSG_Indication_Type v_NasInd;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var CoOrd_GUTIParam_Type v_GUTIFromUTRAN;
    var IRAT_Coordination_MSG v_RxdMsg;
    var octetstring v_OldGUTI;
    var template (omit) MobileIdentity v_MsId_IMSI := f_Imsi2MobileIdentity(px_IMSI_Def);
    var AccessStratumRelease v_Rel; // @sic R5-125738 sic@
    timer t_Wait2 := 2.0; // Wait for TAU complete in step 30A @sic R5-125738 sic@
    v_MsId_IMSI.iei := '23'O;
    //@siclog "Step 1" siclog@
    // Set cell type of Cell A to the "non-Suitable cell"
    // Set cell type of Cell B to the "Serving cell".
    v_CellPowerList := {
      cs_CellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE),
      cs_CellPower(eutra_CellB, tsc_ServingCellRS_EPRE)
    };
    f_EUTRA_SetCellPowerList(v_CellPowerList);
    //@siclog "Step 2" siclog@
    // Check: Does the UE transmit a combined TRACKING AREA UPDATE REQUEST message with Last visited
    //         registered TAI set to TAI-1?
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,
                                               v_RRC_TI,
                                               cr_EstablishmentCause_Any,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                 cdr_TAU_RequestOldGutiLaiTmsiStatus(v_UpdateType,
                                                                                                     f_EUTRA_SecurityKSIasme_Get(),
                                                                                                     cr_DRXparameter_Any ('5C'O) ifpresent, // @sic R5s120210, R5-123117 sic@
                                                                                                     v_AdditionalUpdateType,
                                                                                                     f_GutiParameters2MobileIdentity(omit, v_Guti1_Params),
                                                                                                     v_LAI_1, // @sic R5s110594 sic@
                                                                                                     cr_TAI(v_PLMN, v_TAC1),
                                                                                                     omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;
    v_AdditionalUpdateResult := f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, Combined); // @sic R5s110176 sic@
    //@siclog "Step 3" siclog@
    //The SS sends TRACKING AREA UPDATE ACCEPT message including with LAI set to LAI-2, including
    //         a new TMSI(TMSI-2) as MS identity
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cds_TAU_Accept_T3412(v_UpdateType, // T3412 should be omit
                                                                  omit,
                                                                  v_TaiList2,
                                                                  v_EpsBearerCtxtStatus,
                                                                  v_LAI_2,
                                                                  v_MsId,
                                                                  v_AdditionalUpdateResult,omit)))); // T3412 should be omit
    //@siclog "Step 4" siclog@
    // Check: Does the UE send TRACKING AREA UPDATE COMPLETE message.
    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    //@siclog "Step 5" siclog@
    // The SS releases the RRC connection
    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);
    // The following messages are sent and shall be received on Cell X
    //@siclog "Step 6" siclog@
    // Set cell type of Cell B to the "non-Suitable cell". Set cell type of Cell X to the "Serving cell".
    f_EUTRA_SetCellPower(eutra_CellB, tsc_NonSuitableCellRS_EPRE);
    //UE has new mapped security after TAU procedure on Cell B.
    f_EUTRA_SendAuthParameters(eutra_CellB, UTRAN);
    // @sic R5s110685 sic@
    UTRAN.receive(cmr_IRAT_AuthInfo) -> value v_RxdMsg; // wait for trigger from UTRAN side //Note 1
    v_SecurityParams.AuthParams := v_RxdMsg.Authentication.AuthPS;
    f_EUTRA_Security_Set(v_SecurityParams);
    v_GUTIFromUTRAN := v_RxdMsg.Authentication.GutiParams;
    v_OldGUTI := v_PLMN & bit2oct(v_GUTIFromUTRAN.MME_GroupId) & bit2oct(substr(oct2bit(px_PTMSI_SigDef), 16, 8)) & bit2oct(v_GUTIFromUTRAN.M_TMSI); //Note 1
    // The following messages are sent and shall be received on Cell A
    //@siclog "Step 15" siclog@
    // Set cell type of Cell A to the "Serving cell"
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);
    //@siclog "Step 16 and 17" siclog@
    // The UE performs Cell Reselection from Cell X(UTRAN cell) to Cell A(E-UTRAN cell)
    // Check: Does the UE transmit a combined TRACKING AREA UPDATE REQUEST message with
    //          Last visited registered TAI set to TAI-2?
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                               v_RRC_TI,
                                               cr_EstablishmentCause_Any,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                 cdr_TAU_RequestOldGutiAdditionalGutiOldPtmsiSigLaiTmsiStatus(tsc_EpsUpdate_Combined_IMSIAttach, // @sic R5s110594 sic@
                                                                                                                              f_EUTRA_SecurityKSIasme_Get(),
                                                                                                                              cr_DRXparameter_Any ('5C'O) ifpresent, // @sic R5s120210, R5-123117 sic@
                                                                                                                              v_AdditionalUpdateType,
                                                                                                                              cs_MobileIdentityGuti(omit, v_OldGUTI), // @sic R5s110777 sic@
                                                                                                                              f_GutiParameters2MobileIdentity('50'O, v_Guti1_Params), // @sic R5s110777 sic@
                                                                                                                              v_LAI_4, //  @sic R5s130174 sic@
                                                                                                                              cr_TAI(v_PLMN_B, v_TAC2),
                                                                                                                              omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");
    v_EpsBearerCtxtStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;
    v_AdditionalUpdateResult := f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, Combined); // @sic R5s110176 sic@
    //@siclog "Step 18" siclog@
    // The SS sends TRACKING AREA UPDATE ACCEPT message with LAI set to LAI-1, not including MS identity.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cds_TAU_Accept_T3412(v_UpdateType, // T3412 should be omit
                                                                  omit,
                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN, { v_TAC1 }),
                                                                  v_EpsBearerCtxtStatus,
                                                                  v_LAI_1,
                                                                  omit,
                                                                  v_AdditionalUpdateResult,omit)))); // T3412 should be omit
    //@siclog "Step 19" siclog@
    // The SS releases the RRC connection
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    // The following messages are sent and shall be received on Cell 5
    //@siclog "Step 20" siclog@
    // Set cell type of Cell A to the "non-Suitable cell". Set cell type of Cell 5 to the "Serving cell".
    f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE);
    // @sic R5s110685 sic@
    f_EUTRA_SendAuthParameters(eutra_CellA, UTRAN);
    UTRAN.receive(cmr_IRAT_AuthInfo) -> value v_RxdMsg; // wait for trigger from UTRAN side
    v_SecurityParams.AuthParams := v_RxdMsg.Authentication.AuthPS;
    f_EUTRA_Security_Set(v_SecurityParams);
    v_GUTIFromUTRAN := v_RxdMsg.Authentication.GutiParams;
    v_OldGUTI := v_PLMN & bit2oct(v_GUTIFromUTRAN.MME_GroupId) & bit2oct(substr(oct2bit(px_PTMSI_SigDef), 16, 8)) & bit2oct(v_GUTIFromUTRAN.M_TMSI);
    // The following messages are sent and shall be received on Cell B
    //@siclog "Step 27" siclog@
    // Set cell type of Cell B to the "Serving cell"
    f_EUTRA_SetCellPower(eutra_CellB, tsc_ServingCellRS_EPRE);
    //@siclog "Step 28 and 29" siclog@
    // The UE performs Cell Reselection from Cell 5(UTRAN cell) to Cell B(E-UTRAN cell)
    // Check: Does the UE transmit a combined TRACKING AREA UPDATE REQUEST message with
    //          Last visited registered TAI set to TAI-1?
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellB,
                                               v_RRC_TI,
                                               cr_EstablishmentCause_Any,
                                               cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                 cdr_TAU_RequestOldGutiAdditionalGutiOldPtmsiSigLaiTmsiStatus(tsc_EpsUpdate_Combined_IMSIAttach, // @sic R5s110594 sic@
                                                                                                                              f_EUTRA_SecurityKSIasme_Get(),
                                                                                                                              cr_DRXparameter_Any ('5C'O) ifpresent, // @sic R5s120210, R5-123117 sic@
                                                                                                                              v_AdditionalUpdateType,
                                                                                                                              cs_MobileIdentityGuti(omit, v_OldGUTI), // @sic R5s110777 sic@
                                                                                                                              f_GutiParameters2MobileIdentity('50'O, v_Guti1_Params), // @sic R5s110777 sic@
                                                                                                                              v_LAI_3, //  @sic R5s130174 sic@
                                                                                                                              cr_TAI(v_PLMN, v_TAC1),
                                                                                                                              omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 29");
    v_AdditionalUpdateResult := f_GetAdditionalUpdateResult(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, Combined); // @sic R5s110176 sic@
    //@siclog "Step 30" siclog@
    //  The SS sends TRACKING AREA UPDATE ACCEPT message with LAI set to LAI-2, including
    //         the IMSI as MS identity.
    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellB,
                                 cs_TimingInfo_Now,
                                 cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cds_TAU_Accept_T3412(v_UpdateType, // T3412 should be omit
                                                                  omit,
                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN_B, { v_TAC2 }),
                                                                  v_EpsBearerCtxtStatus,
                                                                  v_LAI_2,
                                                                  v_MsId_IMSI,
                                                                  v_AdditionalUpdateResult,omit)))); // T3412 should be omit
    // EXCEPTION: Step 30a is optional for Release 8 to Release 10 UE implementation, from release 11 onwards, UE shall not send this message.
    //@siclog "Step 30a" siclog@
    // The UE sends a TRACKING AREA UPDATE COMPLETE message.  @sic R5-125738 sic@
    v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease();
    if ((v_Rel == rel8) or (v_Rel == rel9) or (v_Rel == rel10)) {
      t_Wait2.start; // 2 sec considered sufficient
      alt {
        [] SRB.receive(car_SRB1_NasPdu_IND(eutra_CellB,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                      cr_508_TAU_Complete)))
          {
            t_Wait2.stop;
          };
        [] t_Wait2.timeout {};
      }
    }
    //@siclog "Step 31" siclog@
    // The SS releases the RRC connection
    f_EUTRA_RRC_ConnectionRelease(eutra_CellB);
    // The following messages are sent and shall be received on Cell X
    //@siclog "Step 32" siclog@
    // Set cell type of Cell B to the "non-Suitable cell". Set cell type of Cell A to off
    v_CellPowerList := {
      cs_CellPower(eutra_CellA, tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_CellB, tsc_NonSuitableCellRS_EPRE)
    };
    f_EUTRA_SetCellPowerList(v_CellPowerList);   // @sic R5s110726 additional changes for change 2 sic@
    f_EUTRA_SendAuthParameters(eutra_CellB, UTRAN); // @sic R5s110685 sic@
    // wait for trigger from UTRAN side to finish TC
    f_IRAT_WaitForCoOrd_Trigger(UTRAN);
  }// end function fl_TC_9_2_3_2_1a_Body
...


1.2 Change 2
	Object name 
	cds_TAU_Accept_T3412

	Reason for change
	See Change 1

	Summary of change
	Introduced a new template “cds_TAU_Accept_T3412” 

	Source of change
	

	Module
	/LTE_IRAT/9_2/NAS_TrackingArea_UG.ttcn


New:

	...

// new template to pass omit for T3412 (9.2.3.2.1a)                         
  template (value) NAS_DL_Message_Type cds_TAU_Accept_T3412(NAS_AttDetValue_Type p_UpdateValue,
                                                            template (omit) MobileIdentity p_Guti,
                                                            template (omit) TrackingAreaIdList p_TaiList,
                                                            template (omit) EPS_BearerContextStatus p_EPSCtxtStatus,
                                                            template (omit) LocAreaId p_LAI,
                                                            template (omit) MobileIdentity p_MS_Id,
                                                            template (omit) AdditionalUpdateResult p_AdditionalUpdateResult,
                                                            template (omit) GPRS_Timer p_T3412)
  modifies cs_508_TAU_Accept :=
  { /* @status     */
    tRACKING_AREA_UPDATE_ACCEPT := {
      t3412                  := p_T3412
    }
  };
...

